

数式処理：代入・仮定その他

Copyright ©2006 by Shigeto R. Nishitani

▼ 代入, 置換, 仮定に関連したコマンド

▼ *subs*(関係式, *exp1*): 一時的代入

```
> exp1:=x^2-4*x*y+4;
subs(x=2,exp1);
```

$$\text{exp1} := x^2 - 4xy + 4$$

$$8 - 8y \quad (1.1.1.1)$$

```
> subs({x=a+2,y=sin(x)},exp1);
```

$$(a+2)^2 - 4(a+2)\sin(x) + 4 \quad (1.1.1.2)$$

▼ *assume*(関係式): 仮定

```
> sqrt(b^2);
assume(a>0);
sqrt(a^2);
```

$$\sqrt{b^2}$$

$$a \quad (1.1.2.1)$$

▼ *exp1 assuming* 関係式: 一時的仮定

```
> exp1:=x^2-4*x+4;
sqrt(exp1);
```

$$\text{exp1} := x^2 - 4x + 4$$

$$\sqrt{(-2+x)^2} \quad (1.1.3.1)$$

```
> sqrt(exp1) assuming x>2;
```

$$-2 + x \quad (1.1.3.2)$$

▼ *additionally: assume*に加えての仮定

▼ *assign(exp1): solve*で求めた値の確定

```
> x:='x':y:='y':
s1:=solve({x-y+1=0,x+y-2=0},{x,y});
assign(s1);
```

$$s1 := \left\{ y = \frac{3}{2}, x = \frac{1}{2} \right\} \quad (1.1.5.1)$$

```
> x,y;
```

$$\frac{1}{2}, \frac{3}{2} \quad (1.1.5.2)$$

▼ *about: assume*で仮定した内容の確認

```
> about(a);
Originally a, renamed a~:
is assumed to be: RealRange(Open(0),infinity)
```

▼ *anames('user')*: ユーザが定義した変数の確認

```
> anames('user');
```

$$s1, y, x, a \quad (1.1.7.1)$$

▼ *restart,a:='a'*: 値の初期化

▼ 省略操作, その他のコマンド

▼ *||*: 連結作用素, 前後の変数をくっつけて新たな変数とする.

```
> a||1;
a||b;
```

$$a1$$

$$ab \quad (1.2.1.1)$$

```
> for i from 1 to 3 do
a||i:=i^2;
end do;
```

▼ *seq(exp1,i=0..3): for-loop*の単純表記

```
> seq(k,k=4..7);
```

$$4, 5, 6, 7 \quad (1.2.2.1)$$

▼ *map(exp1,i=0..3)*: リスト要素への関数の一括適用

```
> f:=x->exp(-x);
map(f,[0,1,2,3]);
```

$$f := x \rightarrow e^{-x}$$

$$[1, e^{-1}, e^{-2}, e^{-3}] \quad (1.2.3.1)$$

上記の3つを組み合わせると, 効率的に式を扱うことができる.

```
> map(sin,[seq(theta||i,i=0..3)]);
```

$$[\sin(\theta 0), \sin(\theta 1), \sin(\theta 2), \sin(\theta 3)] \quad (1.2.3.2)$$

▼ *add,mul*: 単純な和, 積

```
> add(x^i,i=0..3);
```

$$1 + x + x^2 + x^3 \quad (1.2.4.1)$$

```
> mul(x^i,i=0..3);
```

$$x^6 \quad (1.2.4.2)$$

▼ *sum,product*: 数式に対応した和, 積

```
> add(x^i,i=0..n);
Error, unable to execute add
> sum(x^i,i=0..n);
```

$$\frac{x^{n+1}}{x-1} - \frac{1}{x-1} \quad (1.2.5.1)$$

```
> product(x^i,i=0..n);
```

$$x^{\frac{1}{2}(n+1)^2 - \frac{1}{2}n - \frac{1}{2}} \quad (1.2.5.2)$$

▼ *limit*: 極限

```
> limit(exp(-x),x=infinity);
```

$$0 \quad (1.2.6.1)$$

```
> limit(tan(x),x=Pi/2);
```

$$\text{undefined} \quad (1.2.6.2)$$

```
> limit(tan(x),x=Pi/2,left);
limit(tan(x),x=Pi/2,complex);
```

$$\infty$$

$$-\infty + \infty I \quad (1.2.6.3)$$